徳島LED寺子屋

(2013/06/13　ー　2013/06/27)

Processing で考えよう（1）

第２回：流れの制御とイベント処理

川上　博

2013/06/18

前回の復習：基本図形

```
// file:rev201
// June 18, 2013, H. Kawakami
// TLT lecture on Processing

void setup(){
  size(500,500);
}

void draw(){
  if(mousePressed){
    fill(240,180,0, 50);
  }else{
    fill(0, 180, 200, 50);
  }
  ellipse(mouseX, mouseY, 80, 80);
}
```



3

```
// file:rev202
// June 18, 2013, H. Kawakami
// TLT lecture on Processing

void setup(){
  size(500,500);
  background(200);
}

void draw(){}// 必要です

void mouseDragged(){
  fill(color(random(255),random(255),random(255)));
  ellipse(mouseX, mouseY, 50, 50);
}

void keyPressed(){
  if(key=='e' || key=='E'){
    background(200);
  }
}
```



日本語の注釈を入れる


Preferences
Sketchbook location:
/Users/kawakamihiroshi/Documents/Processing
Browse
Editor font size: 10 (requires restart of Processing)
Use smooth text in editor window (requires restart of Processing)
Enable complex text input (i.e. Japanese, requires restart of Processing)
Increase maximum available memory to 500 MB
Delete previous folder on export
Check for updates on startup
Run sketches on display 1
Launch programs in 32-bit mode 64-bit mode
More preferences can be edited directly in the file
/Users/kawakamihiroshi/Library/Processing/preferences.txt
(edit only when Processing is not running)
OK
Cancel

4, 5章　流れの制御とイベント処理

◎ 変数(４章)　int, float, boolean, char

◎ 繰り返し(４章)　for, while

◎ 判断(５章)　if ~ else, switch() ~ case

◎ イベント(５章)

a) マウス

mouseX, mouseY, pmouseX, pmouseY
mousePressed, mousePressed()

b) キー

key, keyPressed, keyPressed()

◎ map() (５章)

変数と繰り返し(４章)

```
int x, y;
x=150; y=100;
line(x, y, 400, y);
```

使う変数は，まず定義し，
使う前に，値を決めること

```
for(int i=0; i<10; i++){
  line(x, y+30*i, 400, y+30*i);
}
```

```
int i=0;
while(i<10){
  line(x, y+30*i, 400, y+30*i);
  i++;
}
```

繰り返しの流れ図(flow chart)




```
for(int i=50; i<500; i+=50){
  line(i,50,i,100);
}
```




```
int i=50;
while(i<500){
  line(i,50,i,100);
  i+=50;
}
```

網目状に線を引く: lec201




```
// file:lec201
// June 18, 2013, H. Kawakami
// TLT lecture on Processing

int w=400, h=300;
int x=50, y=70;
int m=10, n=10;
float u, v;

u=w/m;
v=h/n;

size(500,500);

rect(x, y, w, h);
stroke(200);
for(int i=0; i < m+1; i++){
  line(x, y+v*i, x+w, y+v*i);
  line(x+u*i, y, x+u*i, y+h);
}
```

升目を塗りつぶす：lec202




```
// file:lec202
// June 18, 2013, H. Kawakami
// TLT lecture on Processing

int x=50, y=50;
int w=40;

size(500,500);

for(int i=0; i<10; i++){
  for(int j=0; j<10; j++){
    fill(color(random(255),random(255),random(255)));
    rect(x+w*i, y+w*j, w, w);
  }
}
```

文字を書く Example 5-19

textAlign(), textSize(), text()




```
int x=100, y=100;
int w=30;

size(500,500);

// matrix array of rects
for(int i=0; i<10; i++){
  for(int j=0; j<10; j++){
    rect(x+w*i, y+w*j, w, w);
  }
}

// label of axes
fill(0);
textSize(18);
textAlign(CENTER, BOTTOM);
for(int i=0; i<=10;i++){
  text(i, x+w*i, 430);
}
for(int i=0; i<=10;i++){
  text(10-i, 80,y+10+w*i);
}
```

keyPressed : lec203




```
// file:lec203
// June 18 2013 by H. Kawakami
// TLT lecture on Processing

void setup() {
  size(500,500);
}

void draw() {
  if(keyPressed){
    if(key=='g'){
      fill(0,255,0);
    }else {
      fill(color(random(255),random(255),random(255)));
    }
  }
  ellipse(mouseX, mouseY, 50, 50);
}
```

if ~ else vs switch ~ case

```
// file:lec204
// June 18, 2013  H. Kawakami
// TLT lecture on Processing

void setup(){
  size(500,500);
  background(255);
}

void draw(){}

void keyPressed(){
  if(key=='r'||key=='R'){
     background(255,0,0);
  }
  if(key=='g'||key=='G'){
     background(0,255,0);
  }
  if(key=='b'||key=='B'){
     background(0,0,255);
  }
}
```

```
// file:lec205
// June 18, 2013  H. Kawakami
// TLT lecture on Processing

void setup(){
  size(500,500);
  background(255);
}

void draw(){}

void keyPressed(){
  switch(key){
    case 'r': case 'R':
      background(255,0,0);
      break;
    case 'g': case 'G':
      background(0,255,0);
      break;
    case 'b': case 'B':
      background(0,0,255);
      break;
    default:
//      background(255);
  }
}
```










```
// file:rev202
// June 18, 2013, H. Kawakami
// TLT lecture on Processing

void setup(){
  size(500,500);
  background(200);
}

void draw(){}// 必要です

void mouseDragged(){
  fill(color(random(255),random(255),random(255)));
  ellipse(mouseX, mouseY, 50, 50);
}

void keyPressed(){
  if(key=='e' || key=='E'){
    background(200);
  }
}
```

データのスケーリング pp.61-63

```
w = map(x, xmin, xmax, wmin, wmax)
```



15

$$w = ax + b$$

where $a=(wmax-wmin)/(xmax-xmin)$;

$b=(wmin*xmax-wmax*xmin)/(xmax-xmin)$



関数のグラフを描く


y
width=500
(500,0)
cubic_function
(0,0)
7.0
y=x^3-x+1
height=500
-5.0
x
-2.0
2.0
(0,360)
(500, 500)

y=x^3-x+1:lec206

```
float plotX1, plotX2, plotY1, plotY2;
float u, v, u0, v0;
float xmin=-2.0, xmax=2.0;
float ymin=-5.0, ymax=7.0;
float h, x, y;
int N=80;
```




```
void setup(){
  size(500,500);
  background(255);

  plotX1=50; plotX2=width-50;
  plotY1=50; plotY2=height-50;

  rectMode(CORNERS);
  stroke(#5679C1);
  rect(plotX1, plotY1, plotX2, plotY2);
  strokeWeight(1);
  v0=map(0, ymax, ymin, plotY1, plotY2);
  line(plotX1, v0, plotX2, v0);
  u0=map(0, xmin, xmax, plotX1, plotX2);
  line(u0, plotY1, u0, plotY2);

  strokeWeight(4);
  h=(xmax-xmin)/N;
  for(int i=0; i<N; i++){
    x=xmin+h*i;
    y=x*x*x-x+1.0;
    u=map(x, xmin, xmax, plotX1, plotX2);
    v=map(y, ymax, ymin, plotY1, plotY2);
    point(u, v);
  }
}
```

2次元写像：2D mapping



```
// file:lec207
// June 18 2013 by H. Kawakami
// TLT lecture on Processing
```

```
float plotX1, plotX2, plotY1, plotY2;
float u, v, u0, v0;
float xmin=-2.0, xmax=2.0;
float ymin=-2.0, ymax=2.0;
float h, x, y;
float a=0.4, b=-1.2;
float x0=0.0, y0=0.0;
int N=80;
```

変数の定義

```
void setup(){
  size(500,500); background(255);

  plotX1=50; plotX2=width-50;
  plotY1=50; plotY2=height-50;

  rectMode(CORNERS); stroke(#5679C1);
  rect(plotX1, plotY1, plotX2, plotY2);

  strokeWeight(1);
  v0=map(0, ymax, ymin, plotY1, plotY2);
  line(plotX1, v0, plotX2, v0);
  u0=map(0, xmin, xmax, plotX1, plotX2);
  line(u0, plotY1, u0, plotY2);

  strokeWeight(2); stroke(0,0,255);
}
```

初期設定

```
void draw(){
  for(int i=0;i<200;i++){
    x=y0+a*x0;
    y=x0*x0+b;
    u=map(x, xmin, xmax, plotX1, plotX2);
    v=map(y, ymax, ymin, plotY1, plotY2);
    point(u, v);
    x0=x;
    y0=y;
  }
}
```

計算とイベント処理